A quick start with *Mathematica*

Steven Tschantz

1/8/09

Notebooks and cells

Mathematica files are called notebooks.

Notebooks record the calculations you do, like an experimental scientist's lab notebook. The result can be edited, organized, and reevaluated as a kind of proof of a calculation. Notebooks can also be annotated with additional text and formatted as reports.

The *Mathematica* frontend is the WYSIWYG editor that you use to prepare notebooks. It communicates with the *Mathematica* kernel, the program that does computation.

A notebook is a vertical sequence of cells and groups of cells, as delimited by brackets at right.

Cells are of various types with distinct formatting and functions. Headings, sections, and text cells allow you to produce a polished report. Input you want evaluated is typed in an input cell. You evaluate the input by pressing shift-return, sending the command to the kernel, with any resulting output placed in a following output cell.

The editing insertion point is indicated by a small vertical bar when typing within a cell, or by a horizontal bar across the window when inserting into the vertical sequence of cells. You may also select the cell brackets to operate on cells or groups of cells, for example to cut or copy whole cells.

Double clicking on a grouping bracket opens or closes a group of cells, allowing a condensed outline view of a document.

If you are at the bottom of the notebook or between cells and start typing, you will start a new input cell. Evaluating the input by pressing shift-return produces an output cell and leaves you after the output ready to start a new input cell. This is the simplest way to start using *Mathematica*.

You can edit and reevaluate an input cell in a notebook by clicking anywhere within the cell, placing the insertion point in the cell for editing, and then pressing shift-return. Alternatively, you can select the cell brackets of one or more cells or groups of cells and press shift-return to reevaluate all of the included input.

Basic computation

Here is an input cell and its corresponding output.

```
2 + 2
```

```
4
```

Mathematica works with symbolic as well as numerical expressions, doing basic simplifications automatically.

```
(x + 2) + (x^2 + 2 x + 3)
```

$5 + 3x + x^2$

Parentheses are used for grouping only. Square brackets [ ] are used to delimit arguments to functions.

```
Sqrt[12.3456]
```

3.51363

Mathematica computes exact symbolic answers when given exact integer input. Numerical answers are computed when approximate numbers are given as decimals, or by the N[ ] function.

```
Sqrt[2]
```

$\sqrt{2}$

```
Sqrt[2.]
```

1.41421

```
N[Sqrt[2.]]
```

1.41421

Capital and lowercase letters are distinguished in *Mathematica* symbols. The built-in constants and functions of *Mathematica* all start with a capital letter. You can use symbols that begin with lowercase without danger of confusion with system functions.

The last result %

In *Mathematica*, the last output value produced can be entered in a new expression as %. However, when you reevaluate input, remember that the value of % is NOT the output immediately above an input cell, unless that was the result that was just produced. This is a handy shortcut, but can become confusing when you have to go back an edit your input. A double percent sign %% denotes the output before last.

```
1 + 2 + 3 + 4
```

10

```
% + 5
```

15

```
% + 6
```

21

```
% + %%
```

36

Assignments

You assign values to symbols using an equals sign. The *Mathematica* kernel remembers each definition until you Clear the symbol or you make a substitute definition.

```
x = 4
```

4

```
x^2 + 2 x + 3
```

27

```
y = 5
```

5

```
x * y
```

20

```
Clear[x]
```

```
x * y
```

5 x

```
Clear[y]
```

Functions are defined by giving the dummy variable(s) followed by an underscore character (specifying a pattern for a rule *Mathematica* should use).

```
f[x_] = Sqrt[x^2 + 1]
```

$\sqrt{1 + x^2}$

```
f[2]
```

$\sqrt{5}$

```
f[2.]
```

2.23607

```
f[y]
```

$\sqrt{1 + y^2}$

```
f[2 x + 1]
```

$\sqrt{1 + (1 + 2 x)^2}$

You can see the current definition of a variable or function using a question mark.

```
? f
```

```
Global`f
```

```
f[x_] = √(1 + x²)
```

```
Clear[f]
```

You may assign names to each step in a computation (instead of using %).

```
sum4 = 1 + 2 + 3 + 4
```

```
10
```

```
sum5 = sum4 + 5
```

```
15
```

```
sum6 = sum5 + 6
```

```
21
```

```
sum5 + sum6
```

```
36
```

Equations are denoted by double equal signs. (It is easy to forget and make an unintended assignment with a single equal sign.)

```
eqn = x^3 - 3 x^2 + 2 x == 0
```

$2x - 3x^2 + x^3 == 0$

```
solns = Solve[eqn, x]
```

```
{{x → 0}, {x → 1}, {x → 2}}
```

Useful functions

```
ExpandAll[(x + 2) (x + 3)]
```

$6 + 5x + x^2$

```
Factor[x^2 + 2 x + 1]
```

$(1 + x)^2$

```
Simplify[x^3 + 3 x^2 + 3 x + 1]
```

$(1 + x)^3$

```
D[x^3 + 2 x^2 + 3 x, x]
```

$3 + 4x + 3x^2$

```
Integrate[x^3 + 2 x^2 + 3 x, x]
```

$$\frac{3x^2}{2} + \frac{2x^3}{3} + \frac{x^4}{4}$$

```
Integrate[x^3 + 2 x^2 + 3 x, {x, 0, 1}]
```

$$\frac{29}{12}$$

```
Solve[x^2 - 2 x - 3 == 0, x]
```

```
{{x → -1}, {x → 3}}
```

```
Plot[x^2 - 2 x - 3, {x, -2, 4}]
```



Lists, tables, rules, and substitution.

A list is delimited by curly braces { } with elements separated by commas. An ordered pair, say for the coordinates of a point, is given as {x,y}.

```
list1 = {1, 2, 3, 4, 5}
```

```
{1, 2, 3, 4, 5}
```

Double square brackets [[ ]] are used to access the i-th element of a list.

```
list1[[4]]
```

```
4
```

The Plot command takes as second argument a list consisting of the independent variable and the minimum and maximum values of the variable to plot.

```
Plot[Sin[x], {x, 0, 2 Pi}]
```



You can produce a list of values of some expression for a range of values of a variable using Table.

```
Table[i^2, {i, 3, 5}]
```

```
{9, 16, 25}
```

A matrix is represented as a list of its rows, each row given as a list of elements.

```
matrix1 = {{1, 2}, {3, 4}}
```

```
{{1, 2}, {3, 4}}
```

To display this as a matrix you use MatrixForm

```
MatrixForm[matrix1]
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$$

```
Det[matrix1]
```

```
-2
```

A period is used for the dot product of vectors or multiplication by a matrix.

```
vector1 = {5, 6}
```

```
{5, 6}
```

```
vector1.vector1
```

```
61
```

```
matrix1.vector1
```

```
{17, 39}
```

The Solve function will take a system of equations given as a list and solve for the variables given in a list.

```
solns = Solve[{2 x + 3 y == 7, x - y == 1}, {x, y}]
```

```
{{x → 2, y → 1}}
```

Solve returns a list of solutin sets. Each solution set is a list of rules giving values to the variables. A rule is of the form var -> val (the arrow is typed as minus greater but converted to an arrow). To use a rule to substitute values of variables in an expression you use ReplaceAll which has the handy shorthand infix notation /. (forward slash and period). If you substitute a list of rules you get one result. If you substitute a list of lists of rules you get one result for each list of rules. Thus if you want to use the first solution set to substitute in an expression you would use

```
x + y /. solns[[1]]
```

```
3
```

```
solns = Solve[x^2 - 3 x + 2 == 0, x]
```

```
{{x → 1}, {x → 2}}
```

```
x^2 /. solns
```

```
{1, 4}
```

```
x^2 /. solns[[2]]
```

```
4
```

Help

You can access all of the documentation for *Mathematica* from the Help menu Documentation Center. If you want to know about a specific function, you can highlight the function and select Find Selected Function from the Help menu. If you go to the Help menu Documentation Center, there is a link to First Five Minutes with *Mathematica* on the lower right side of the page.